"Lunch on the nature" - 2

More detailed model of the game

First of all we introduce the enumerable type to characterize current state of the game. Success and the Lunch correspond to the two possible results. Working means that the game is not finished yet.
enum Situation

  Working
  Success
  Lunch
Clearly, we have two types of individuals in our game.
enum PersonType
  Cannibal
  Preacher
We don't keep any additional information about persons beside the type.
class Person

  const Kind as PersonType  
The boat is characterized by its residents and by the current position. We introduce two auxiliary functions to check whether the boat is empty and to calculate the difference between positive part of the residents (preachers) and its negative part (cannibals).
class Boat
  constraint Size(Travelers) <=2
  var Travelers as Set of Person  
  var CurrentPosition as RiverSide

  function IsNotEmpty() as Boolean

    return Size(Travelers) in {1,2}

  function Score() as Integer

    let positive = { person | person in Travelers where person.Kind = Preacher }

    let negative = { person | person in Travelers where person.Kind = Cannibal }    
    return Size(positive) – Size(negative)
Now, there are three actions that involve boat: moving a person in, moving somebody out, and moving the boat to another side of the river.
class Boat

  procedure MoveTo(OtherSide as RiverSide)

    require Size(Travelers) in {1,2}
    CurrentPosition := OtherSide
  procedure MoveIn(person as Person)

    require person in CurrentPosition.Walkers

    require Size(Travelers) in {0,1}
    remove person from CurrentPosition.Walkers

    add person to Travelers

  procedure MoveOut(person as Person)

    require person in Travelers

    remove person from Travelers

    add person to CurrentPosition.Walkers
The riverside is characterized by the set of the relaxing tourists. Here we also need to compute the score.
class RiverSide

  var Walkers as Set of Person
  function Score() as Integer

    let positive = { person | person in Walkers where person.Kind = Preacher }

    let negative = { person | person in Walkers where person.Kind = Cannibal }    
    return Size(positive) – Size(negative)
Now we define the state of the game. It is characterized by the two riversides, the boat and the overall attitude.
const empty = {} as Set of Person
var LeftSide  as RiverSide = new RiverSide(empty) 

var RightSide as RiverSide = new RiverSide ( { new Person(Cannibal) | i in {1..3} }

                                           + { new Person(Preacher) | i in {1..3} } )

var TheBoat as Boat = new Boat( empty, RightSide ) 

var Attitude as Situation = Working

The victory conditions are formalized by the following two functions.
function ReadyForLunch() as Boolean

  return Attitude = Working 

    and TheBoat.Score() + TheBoat.CurrentPosition.Score() < 0

function Victory() as Boolean 

  return Attitude = Working 

    and RightSide.Walkers = empty 

    and TheBoat.CurrentPosition = LeftSide
function ShowIsGoingOn() as Boolean

  return not ( Victory() or ReadyForLunch() )
The transitions include the boat actions and two finalizing steps: the lunch (cannibals win) and the victory.
procedure MoveBoat()
  require ShowIsGoingOn()
  require TheBoat.IsNotEmpty()
  if TheBoat.CurrentPosition = LeftSide then
    TheBoat.CurrentPosition := RightSide
  else
    TheBoat.CurrentPosition := LeftSide
procedure MoveCharacterIn(character as PersonType)
  require ShowIsGoingOn()
  require exists p in TheBoat.CurrentPosition.Walkers where p.Kind = character

  choose person in TheBoat.CurrentPosition.Walkers where person.Kind = character

    TheBoat.MoveIn(person)

procedure MoveCharacterOut(character as PersonType)

  require ShowIsGoingOn()
  require exists p in TheBoat.Travelers where p.Kind = character

  choose person in TheBoat.Travelers where person.Kind = character

    TheBoat.MoveOut(person)

procedure Terminate()

  require ReadyForLunch() or Victory()

  if ReadyForLunch() then
    Attitude := Lunch
  else
    Attitude := Success
Main()
  step until Attitude ne Working
      ShowState()
      AnyAction()
  step
      ShowState() 
      WriteLine(Attitude)
procedure AnyAction()

  try RandomAction()

  catch e: skip 
procedure RandomAction()
  let actions = {"MoveBoat", "MoveCannibalIn", "MoveCannibalOut",

                   "MovePreacherIn", "MovePreacherOut", "Terminate"}

  choose action in actions
    WriteLine(action)
    match action

        "MoveBoat"       : MoveBoat()
        "MoveCannibalIn" : MoveCharacterIn(Cannibal)
        "MoveCannibalOut": MoveCharacterOut(Cannibal)
        "MovePreacherIn" : MoveCharacterIn(Preacher)
        "MovePreacherOut": MoveCharacterOut(Preacher)
        "Terminate"      : Terminate()
procedure ShowState()

  let left  = [ p.Kind | p in ToSeq(LeftSide.Walkers) ]
  let right = [ p.Kind | p in ToSeq(RightSide.Walkers) ] 

  let boat =  [ p.Kind | p in ToSeq(TheBoat.Travelers) ]

  let boatside = if TheBoat.CurrentPosition = LeftSide then "left" else "right"

  WriteLine("=== Current state of the game ===")
  WriteLine("  Left riverside: " + ToString(left))

  WriteLine("  Right riverside: " + ToString(right))

  WriteLine("  The boat is on the " + boatside + ":" + ToString(boat)) 
function GlobalState() as String
  if Attitude = Lunch then 

    return "Lunch"

  elseif Attitude = Success then 

    return "Victory"

  else 

    let left  = ToString( Numbers( Size(LeftSide.Walkers), LeftSide.Score() ) )

    let right = ToString( Numbers( Size(RightSide.Walkers), RightSide.Score() ) ) 

    let boat =  ToString( Numbers( Size(TheBoat.Travelers), TheBoat.Score() ) )

    let boatside = if TheBoat.CurrentPosition = LeftSide then "left" else "right"

    return left + right + boat + boatside
function Numbers(total as Integer, score as Integer) as (Integer, Integer)

   return ( (total+ score) / 2, (total - score) / 2)

